
*=======TRANSIENT_OSCILLATOR_PHASE_NOISE_SPECTRUM=============

Adding a piece wise linear noise source to a free running oscillator 
can show how the noise maps to jitter or PM or to a spectrum. 

In this case a 2second/1kHz BW random PWL file is created.

OSCILLATOR_WITH_PWL_Noise
*                R1
*    _____/\  /\  /\__
*    |      \/  \/   _|_
* VFB|               ///
*    |___/\  /\  /\_____________
*    |     \/  \/       _|_     |
*   _|_           R2    ___ C0  |   ___
*  /_  \ BPWL            |      |__|OUT|
* // \  \               _|_200uF|  |___|
* \   \//               ///     |
*  \___/    INP |\              |
*    |__________|+\             |
*               |  \_/\  /\  /\_|
*    C1    INN  |  /   \/  \/   |
*       ________|-/     R0 1 Ohm|
*     _|_  |    |/ BAMP         |
*     ___  |                    |
*      |   |     R3             |
*     _|_  |__/\  /\  /\________|
*     ///       \/  \/
*
* VpwlT OUT 0        PWL( + 0.0005 0.988835 +.....
*=========Include_Noise_File=================================================
.include  PWL_File.inc
Rload     OUT    0      1k
BAMP      OUT1   0      V =    9.9*tanh( (V(INP)-V(INN))*10) 
R0        OUT1   OUT2   1
C0        OUT2   0      500u
R1        VFB    0      1K
R2        VFB    OUT2   1K
R3        INN    OUT2   2.49K
BPWL      INP    VFB    V =    .1*V(OUT)
C1        INN    0      9.415u IC= .1
*=========Run_Simulation=================================================
*TRAN     TSTEP  TSTOP  TSTART TMAX      ?UIC?
.tran     5u     2      0      5u        UIC
.control
run
set       pensize =     2
plot      out2   inp    inn
*=======================================================================

Noise from BPWL is applied to one of the inputs of the amplifier BAMP.
Amplifier BAMP is clipping its output at +/- 10Volts. 

Capacitor C1 is toggling between +/- 5volts. Noise source BPWL is
messing up the timing. So in this case C1 has been adjusted to produce 
40 whole cycles within 2 seconds. 

Noise source BPWL is receiving a random 1VRMS signal with a bandwidth of
1KHz. This has been scaled down to +/- 100mV rms.



The Noise source BPWL can be seen on the inp input. But its PM effects are 
hard to see. A spectrum can show the details.

*=========Find_Spectrum=================================================
linearize 
set       specwindow=   "none" 
spec      .5    1k      .5    v(out2)
plot      mag(out2)  loglog
plot      mag(out2)  loglog xlimit 10 100 ylimit  .001 100
*=========Wrap_Up============================================
.endc
.end
*=======================================================================

       
The actual noise is appearing close to the peak signals at 20Hz 
and 60Hz and 100Hz, etc.  Since white noise was applied,
the slope of the adjacent noise sidebands around 20Hz 
drop in power to the  1/f^2 as they get farther away from
20Hz. The same is true for all the odd harmonics.

But it is also possible to get the same plot by doing a jitter
plot on this oscillator, and then creating a PWL file which 
will be used to phase modulate a 20Hz signal. This will produce
a spectrum which is very close. 

And it is possible to map how the noise gets around to producing
its sidebands at its magnitude and at its frequency.

===========Full_Netlist_For_Copy_Paste=======================
OSCILLATOR_WITH_PWL_Noise
*                R1
*    _____/\  /\  /\__
*    |      \/  \/   _|_
* VFB|               ///
*    |___/\  /\  /\_____________



*    |     \/  \/       _|_     |
*   _|_           R2    ___ C0  |   ___
*  /_  \ BPWL            |      |__|OUT|
* // \  \               _|_200uF|  |___|
* \   \//               ///     |
*  \___/    INP |\              |
*    |__________|+\             |
*               |  \_/\  /\  /\_|
*    C1    INN  |  /   \/  \/   |
*       ________|-/     R0 1 Ohm|
*     _|_  |    |/ BAMP         |
*     ___  |                    |
*      |   |     R3             |
*     _|_  |__/\  /\  /\________|
*     ///       \/  \/
*
* VpwlT OUT 0        PWL( + 0.0005 0.988835 +.....
*=========Include_Noise_File=================================================
.include  PWL_File.inc
Rload     OUT    0      1k
BAMP      OUT1   0      V =    9.9*tanh( (V(INP)-V(INN))*10) 
R0        OUT1   OUT2   1
C0        OUT2   0      500u
R1        VFB    0      1K
R2        VFB    OUT2   1K
R3        INN    OUT2   2.49K
BPWL      INP    VFB    V =    .1*V(OUT)
C1        INN    0      9.415u IC= .1
*=========Run_Simulation=================================================
*TRAN     TSTEP  TSTOP  TSTART TMAX      ?UIC?
.tran     5u     2      0      5u        UIC
.control
run
set       pensize =     2
plot      out2   inp    inn
*=========Find_Spectrum=================================================
linearize 
set       specwindow=   "none" 
spec      .5    1k      .5    v(out2)
plot      mag(out2)  loglog
plot      mag(out2)  loglog xlimit 10 100 ylimit  .001 100
*=========Wrap_Up============================================
.endc
.end

